
Change

	Function name
	F_22_4_6_NBIOT()

	Reason for change
	When paging the UE using Direct Indication at step 3, the 3rd paramter to function f_NBIOT_UE_PageSysinfoMod() should be set to “true_” instead of “omit” so that the function calculates PF/PO correctly

	Summary of change
	Change 3rd paramter of _NBIOT_UE_PageSysinfoMod() from “omit” to “-“ (= “true_”)

	TTCN module
	NBIOT_Paging.ttcn

	MCC160 Comment
	Accepted in principle but implemented differently.  Also called new function f_NBIOT_UE_PageSysinfoMod_eDRX from 22.4.1 step 14.
Similar changes also implemented for EUTRA sister functions.


Before change

    function f_TC_22_4_6_NBIOT() runs on NBIOT_PTC

  {

    var  SystemInformationBlockType1_NB v_SIB1;

    var SystemInformationBlockType2_NB_r13 v_SIB2;

    var SubFrameTiming_Type v_TimeOfPaging;

    var NPRACH_ConfigSIB_NB_r13 v_NPRACH_Config;

    f_NBIOT_Init(c1);

    // set systemInfoValueTag in SIB1 to start at 0

    v_SIB1 := f_NBIOT_CellInfo_GetSIB1(nbiot_Cell1);

    v_SIB1.systemInfoValueTagList_r13[0] := 0; // @sic R5-170571 sic@

    f_NBIOT_CellInfo_SetSIB1(nbiot_Cell1, v_SIB1);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE3_NB_IDLEUPDATED);

    f_NBIOT_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    f_Delay(5.0);

    //@siclog "Step 2" siclog@

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);

    v_SIB1.systemInfoValueTagList_r13[0] := v_SIB1.systemInfoValueTagList_r13[0] + 1;

    v_NPRACH_Config := v_SIB2.radioResourceConfigCommon_r13.nprach_Config_r13;

    v_NPRACH_Config.nprach_CP_Length_r13  := us266dot7; // change value

    v_SIB2.radioResourceConfigCommon_r13.nprach_Config_r13 := v_NPRACH_Config; // update global variable

    // re-configure SS with new value

    f_NBIOT_SS_CommonCellConfig(nbiot_Cell1,

                                cads_NB_CellConfig_UpdatePRACH_Info_REQ(nbiot_Cell1,

                                                                        cs_TimingInfo_Now,

                                                                        -,

                                                                        v_NPRACH_Config));

    // Store new values of SIBs

    f_NBIOT_CellInfo_SetSIB1(nbiot_Cell1, v_SIB1);

    f_NBIOT_CellInfo_SetSIB2(nbiot_Cell1, v_SIB2);

    // Send modified SIBs

    v_TimeOfPaging := f_NBIOT_ModifySysinfo(nbiot_Cell1, false); // Send Paging message in next step

    //@siclog "Step 3" siclog@

    // Send Paging message with Direct Indication

    f_NBIOT_UE_PageSysinfoMod(nbiot_Cell1, v_TimeOfPaging, omit, omit, '00000001'B); // Direct Indication SystemInfoModification

    //@siclog "Step 4" siclog@

    f_Delay(90.0);

    //@siclog "Step 5" siclog@

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    //@siclog "Step 6" siclog@

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell1, CONTROL_PLANE, N1_IDLE);

  }
After change

    function f_TC_22_4_6_NBIOT() runs on NBIOT_PTC

  {

    var  SystemInformationBlockType1_NB v_SIB1;

    var SystemInformationBlockType2_NB_r13 v_SIB2;

    var SubFrameTiming_Type v_TimeOfPaging;

    var NPRACH_ConfigSIB_NB_r13 v_NPRACH_Config;

    f_NBIOT_Init(c1);

    // set systemInfoValueTag in SIB1 to start at 0

    v_SIB1 := f_NBIOT_CellInfo_GetSIB1(nbiot_Cell1);

    v_SIB1.systemInfoValueTagList_r13[0] := 0; // @sic R5-170571 sic@

    f_NBIOT_CellInfo_SetSIB1(nbiot_Cell1, v_SIB1);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE3_NB_IDLEUPDATED);

    f_NBIOT_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    f_Delay(5.0);

    //@siclog "Step 2" siclog@

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);

    v_SIB1.systemInfoValueTagList_r13[0] := v_SIB1.systemInfoValueTagList_r13[0] + 1;

    v_NPRACH_Config := v_SIB2.radioResourceConfigCommon_r13.nprach_Config_r13;

    v_NPRACH_Config.nprach_CP_Length_r13  := us266dot7; // change value

    v_SIB2.radioResourceConfigCommon_r13.nprach_Config_r13 := v_NPRACH_Config; // update global variable

    // re-configure SS with new value

    f_NBIOT_SS_CommonCellConfig(nbiot_Cell1,

                                cads_NB_CellConfig_UpdatePRACH_Info_REQ(nbiot_Cell1,

                                                                        cs_TimingInfo_Now,

                                                                        -,

                                                                        v_NPRACH_Config));

    // Store new values of SIBs

    f_NBIOT_CellInfo_SetSIB1(nbiot_Cell1, v_SIB1);

    f_NBIOT_CellInfo_SetSIB2(nbiot_Cell1, v_SIB2);

    // Send modified SIBs

    v_TimeOfPaging := f_NBIOT_ModifySysinfo(nbiot_Cell1, false); // Send Paging message in next step

    //@siclog "Step 3" siclog@

    // Send Paging message with Direct Indication

    f_NBIOT_UE_PageSysinfoMod(nbiot_Cell1, v_TimeOfPaging, -, omit, '00000001'B); // Direct Indication SystemInfoModification

    //@siclog "Step 4" siclog@

    f_Delay(90.0);

    //@siclog "Step 5" siclog@

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    //@siclog "Step 6" siclog@

    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell1, CONTROL_PLANE, N1_IDLE);

  }
MCC160 Implementation:
  function f_NBIOT_UE_PageSysinfoMod(NBIOT_CellId_Type p_CellId,

                                     SubFrameTiming_Type  p_StartSFN,

                                     template (omit)  NB_Paging_SystemInfoModification_Type p_SystemInfoModification := true_,

                                     template (omit)  NB_Paging_SystemInfoModification_eDRX_Type p_SystemInfoMod_eDRX := omit,

                                     template (omit)  B8_Type p_DirectIndication := omit)

    runs on NBIOT_PTC

  {

    var PCCH_Config_NB_r13 v_SI_Pcch_Configuration := f_NBIOT_CellInfo_GetPCCH_ConfigInSYSINFO(p_CellId);  // Derive DRX params from SYS Info

    var SubFrameTimingList_Type v_PoList;

    var template (value) SubFrameTiming_Type v_SubFrameTiming;

    if (isvalue ( p_SystemInfoMod_eDRX)) { // eDRX @sic R5s170460 sic@

      // When UE is in IDLE mode: page UE at PF/PO calculated as per 36.304 cl 7

      // Calculate next Paging Occasion

      v_PoList := f_NBIOT_Calculate_PF_PO(p_StartSFN.SFN.Number, p_StartSFN.HSFN.Number, v_SI_Pcch_Configuration.defaultPagingCycle_r13, v_SI_Pcch_Configuration.nB_r13, px_IMSI_Def);

      // Calculate the subframe offset list

      // Page UE in each pagingCycle

      // The first offset is 0: the first reference time is the PagingOccassion

      v_SubFrameTiming := v_PoList[0];

    } else { // non-eDRX

      v_SubFrameTiming := p_StartSFN; // already calculated the PO before calling this function

    }

    // Send Paging indicating SysinfoModification

    if (isvalue(p_DirectIndication)) {

      SYS.send(cas_NB_CellConfig_PagingSysinfoMod_DirectIndication_REQ(p_CellId, v_SubFrameTiming, p_DirectIndication));

    } else {

      SYS.send(cas_NB_CellConfig_PagingSysinfoMod_REQ(p_CellId, v_SubFrameTiming, omit, p_SystemInfoModification, p_SystemInfoMod_eDRX));

    }

  }
  function f_NBIOT_ModifySysinfo_eDRX(NBIOT_CellId_Type p_CellId,

                                      integer p_PTWLength_Index,

                                      integer p_EDRXValue_Index,

                                      template (omit)  NB_Paging_SystemInfoModification_eDRX_Type p_SystemInfoMod_eDRX := omit) runs on NBIOT_PTC

  {


    var SubFrameTiming_Type v_Timing;

    var SubFrameTiming_Type v_StartNextModPeriod;

    var integer v_ModificationPeriodFrames := f_NB_CalculateModificationPeriod(p_CellId);

    //Get the current SFN and subframe

    v_Timing := f_NBIOT_GetCurrentTiming(p_CellId);


    f_NBIOT_UE_PageSysinfoMod _eDRX ( p_CellId,  p_PTWLength_Index, p_EDRXValue_Index, p_SystemInfoMod_eDRX);

    // Now calculate next modification period to start sending the new SIBs

    v_StartNextModPeriod.HSFN.Number := (v_Timing.HSFN.Number + (v_Timing.SFN.Number + v_ModificationPeriodFrames)/1024) mod 1024;

    v_StartNextModPeriod.SFN.Number := (v_Timing.SFN.Number + v_ModificationPeriodFrames) mod 1024;

    f_NBIOT_SS_ConfigureSysinfo(p_CellId, true, cs_TimingInfo_SFN(v_StartNextModPeriod.SFN.Number, v_StartNextModPeriod.HSFN.Number));

  }
  /*

   * @desc      Trigger SS to send paging request to indicate sysinfo modification to UE for eDRX

   * @param     p_CellId

   * @param     p_PTWLength_Index

   * @param     p_EDRXValue_Index

   * @param     p_SystemInfoMod_eDRX (default value: true_)

   * @param     p_DirectIndication  (default value: omit)

   */

  function f_NBIOT_UE_PageSysinfoMod_eDRX(NBIOT_CellId_Type p_CellId,

                                          integer p_PTWLength_Index,

                                          integer p_EDRXValue_Index,

                                          template (omit)  NB_Paging_SystemInfoModification_eDRX_Type p_SystemInfoMod_eDRX := true_,

                                          template (omit)  B8_Type p_DirectIndication := omit)

    runs on NBIOT_PTC

 {

    var SubFrameTimingList_Type v_PoList;

    v_PoList := f_NBIOT_PO_List_eDRX (p_CellId, p_PTWLength_Index, p_EDRXValue_Index);

    f_NBIOT_UE_PageSysinfoMod( p_CellId, v_PoList[0], omit, p_SystemInfoMod_eDRX, p_DirectIndication);

 }  
  function f_EUTRA_ModifySysinfo_eDRX(EUTRA_CellId_Type                 p_CellId,

                                      integer                           p_PTWLength_Index,

                                      integer                           p_EDRXValue_Index,

                                      template (omit) Paging_v890_IEs   p_Paging_v890_IEs) runs on EUTRA_PTC

  {


    var SubFrameTiming_Type v_Timing;

    var SubFrameTiming_Type v_StartNextModPeriod;

    var integer v_ModificationPeriodFrames := f_CalculateModificationPeriod (p_CellId);

    //Get the current SFN and subframe

    v_Timing := f_EUTRA_GetCurrentTiming(p_CellId);


    f_EUTRA_UE_PageSysinfoMod_eDRX( p_CellId,  p_PTWLength_Index, p_EDRXValue_Index, p_Paging_v890_IEs);

    // Now calculate next modification period to start sending the new SIBs

    v_StartNextModPeriod.HSFN.Number := (v_Timing.HSFN.Number + (v_Timing.SFN.Number + v_ModificationPeriodFrames)/1024) mod 1024;

    v_StartNextModPeriod.SFN.Number := (v_Timing.SFN.Number + v_ModificationPeriodFrames) mod 1024;

    f_EUTRA_SS_ConfigureSysinfo(p_CellId, cs_TimingInfo_SFN(v_StartNextModPeriod.SFN.Number, v_StartNextModPeriod.HSFN.Number));

  }

  function f_EUTRA_UE_PageSysinfoMod(EUTRA_CellId_Type p_CellId,

                                     SubFrameTiming_Type  p_StartSFN, //@sic R5w160204r1 sic@

                                     EUTRA_RRC_STATE_Type p_UE_State := RRC_IDLE, // @sic R5-110803 change 4: add new parameter sic@

                                     template (omit)  Paging_SystemInfoModification_Type p_SystemInfoModification := true_,

                                     template (omit)  Paging_ETWS_Indication p_EtwsIndication := omit,

                                     template (omit)  Paging_v890_IEs p_CMAS_Paging := omit,

                                     boolean p_DirectIndication := false,

                                     boolean p_EDRX := false)  // @sic R5-121854 sic@

    runs on EUTRA_PTC

  {

    // @sic R5-110803 change 4 sic@

    // @sic R5-169033 eMTC added p_DirectIndication sic@

    var PCCH_Config v_SI_Pcch_Configuration := f_EUTRA_CellInfo_GetPCCH_ConfigInSYSINFO(p_CellId);  // Derive DRX params from SYS Info

    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(p_CellId);      // Derive FDD or TDD mode from Cell-configuration

    var SubFrameTimingList_Type v_PoList;

    var template (omit) SubframeOffsetList_Type v_SubframeOffsetList:=omit; //@sic R5-167569: for sysinfo-BR page UE only once sic@

    var template (value) SubFrameTiming_Type v_SubFrameTiming;

    var integer v_T := f_EUTRA_GetPagingCycleValue(v_SI_Pcch_Configuration.defaultPagingCycle);

    var EUTRA_CellSysInfo_Type  v_Sysinfo := f_EUTRA_CellInfo_GetSYSINFO (p_CellId);

    var integer v_ModPeriodcoef;

    var integer v_Nb;

    var integer v_Ns;

    var integer i;

    var integer k;

    var SysinfoType_Type v_SysinfoType := f_EUTRA_CellInfo_GetSysinfoType(p_CellId);

    // Get the modification period coefficient from sysinfo

    select (v_Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon.bcch_Config.modificationPeriodCoeff){

      case (n2) {v_ModPeriodcoef:=2}

      case (n4) {v_ModPeriodcoef:=4}

      case (n8) {v_ModPeriodcoef:=8}

      case else{ FatalError (__FILE__, __LINE__, "invalid modification period coefficient");}

    }

    select (p_UE_State) {

      case (RRC_IDLE) {

        if (p_EDRX) {  // @sic R5s170792 sic@

          // already calculated the PO before calling this function

          v_SubFrameTiming := p_StartSFN;

        } else {        // When UE is in IDLE mode: page UE at PF/PO calculated as per 36.304 cl 7

        // Paging is sent at PF/PO during the modification Period: paging is sent modificationPeriodCoef times

        // Calculate next Paging Occasion

        v_PoList := f_EUTRA_Calculate_PF_PO(p_StartSFN.SFN.Number, p_StartSFN.HSFN.Number, v_SI_Pcch_Configuration.defaultPagingCycle, v_SI_Pcch_Configuration.nB, px_IMSI_Def, v_EUTRA_FDD_TDD_Mode, v_SysinfoType);

        // Calculate the subframe offset list

        // Page UE in each pagingCycle

        // The first offset is 0: the first reference time is the PagingOccassion

        v_SubFrameTiming := v_PoList[0];
       }
        for (i := 0; i < v_ModPeriodcoef; i := i + 1) {

          v_SubframeOffsetList[i] := i*v_T*10;

        }

      }

      case (RRC_CONNECTED) {

        // When UE is in CONNECTED mode, page UE during the complete modification period

        // Calculate integer type nB based on enumerated type p_Nb and integer type T

        select (v_SI_Pcch_Configuration.nB) {

          case (fourT) { v_Nb := 4*v_T; }

          case (twoT) { v_Nb := 2*v_T; }

          case (oneT) { v_Nb := v_T; }

          case (halfT) { v_Nb := v_T/2; }

          case (quarterT) { v_Nb := v_T/4; }

          case (oneEighthT) { v_Nb := v_T/8; }

          case (oneSixteenthT) { v_Nb := v_T/16; }

          case (oneThirtySecondT) { v_Nb := v_T/32; }

          case else { FatalError(__FILE__, __LINE__, "Provided value for nB is unrecognizable"); }

        }

        // Derive N and Ns based on values on nB and T

        if (1 >= (v_Nb/v_T)) {

          v_Ns := 1;

        } else {

          v_Ns := v_Nb/v_T;

        }

        // When UE supports ETWS or CMAS, then Paging message is send during a Paging cycle only

        // Acc. 36.331 cl 5.2.1.3

        if (pc_ETWS_message or pc_CMAS_Message) { // @sic R5-121854 sic@

          v_ModPeriodcoef := 1;

        }

        // Calculate the subframe offset list

        // Page UE at each SFN of the modification period or pagyingCycle for ETWS/CMAS capable UE

        select (v_Ns) {

          case (1) {

            // Calculate the first Paging Frame and Paging Occassion

            if (v_EUTRA_FDD_TDD_Mode == FDD) {

              v_SubFrameTiming := cs_SubFrameTiming(p_StartSFN.SFN.Number, 9, p_StartSFN.HSFN.Number);

            } else {  // TDD

              v_SubFrameTiming := cs_SubFrameTiming(p_StartSFN.SFN.Number, 0, p_StartSFN.HSFN.Number);

            };

            // Calculate the offset list

            // for FDD page at subframes 9 (offset 10)

            // for TDD page at subframes 0 (offset 10)

            // The first offset is 0: the first reference time is the PagingOccassion

            for (i := 0; i < (v_ModPeriodcoef*v_T) ; i := i + 1) {

              v_SubframeOffsetList[i] := i*10;

            }

          }

          case (2) {

            // Calculate the first Paging Frame and Paging Occasion

            if (v_EUTRA_FDD_TDD_Mode == FDD) {

              v_SubFrameTiming := cs_SubFrameTiming(p_StartSFN.SFN.Number, 4, p_StartSFN.HSFN.Number);

            } else {  // TDD

              v_SubFrameTiming := cs_SubFrameTiming(p_StartSFN.SFN.Number, 0, p_StartSFN.HSFN.Number);

            };

            // Calculate the offset acc to 36.304 cl 7.2

            // for FDD page at subframes 4,9 (offset 5)

            // for TDD page at subframes 0,5 (offset 5)

            // The first offset is 0: the first reference time is the PagingOccassion

            for (i := 0; i < (v_ModPeriodcoef*v_T*2); i := i + 1) {

              v_SubframeOffsetList[i] := i*5;

            }

          }

          case (4) {

            // Calculate the first Paging Frame and Paging Occasion

            v_SubFrameTiming := cs_SubFrameTiming(p_StartSFN.SFN.Number, 0, p_StartSFN.HSFN.Number);

            // Calculate all subframe offsets acc  to 36.304 cl 7.2

            // The first offset is 0: the first reference time is the PagingOccasion

            k:=0;  // Used to calculate the frame offset

            for (i := 1; i < (v_ModPeriodcoef*v_T*4); i := i + 4) {

              // i is used to point the elements in the list

              if (v_EUTRA_FDD_TDD_Mode == FDD) {

                // For FDD page at subframes 0,4,5,9

                v_SubframeOffsetList[i] := k*10;

                v_SubframeOffsetList[i+1] := k*10+4;

                v_SubframeOffsetList[i+2] := k*10+5;

                v_SubframeOffsetList[i+3] := k*10+9;

              } else {

                // For TDD page at subframes 0,1,5,6

                v_SubframeOffsetList[i] := k*10;

                v_SubframeOffsetList[i+1] := k*10+1;

                v_SubframeOffsetList[i+2] := k*10+5;

                v_SubframeOffsetList[i+3] := k*10+6;

              }

              k := k+1;

            }

          }

          case else {FatalError(__FILE__, __LINE__, "Ns has unrecognisable value");}

        }

      }

      case else {FatalError(__FILE__, __LINE__, "UE State has unrecognisable value");}

    }

    if (v_SysinfoType == sysinfoAndBR) {

        //@sic R5-157569 sic@

        //For sysinfo-BR: page UE only once at v_SubFrameTiming

        v_SubframeOffsetList := omit;

    };

    // Send Paging indicating SysinfoModification or Direct indictaion information indication SysinfoModification

    if (not(p_DirectIndication)){

      SYS.send(cas_CellConfig_PagingSysinfoMod_REQ(p_CellId, v_SubFrameTiming, omit, v_SubframeOffsetList, p_SystemInfoModification, p_EtwsIndication, p_CMAS_Paging));

    } else {

      // @sic R5-169033 eMTC added p_DirectIndication sic@

      SYS.send(cas_CellConfig_DirectIndication_REQ(p_CellId, v_SubFrameTiming, '00000001'B, v_SubframeOffsetList));

    }

    /* @sic R5s100135 CNF removed sic@ */

  }
  function f_EUTRA_UE_PageSysinfoMod_eDRX(EUTRA_CellId_Type p_CellId,

                                          integer p_PTWLength_Index,

                                          integer p_EDRXValue_Index,

                                          template (omit) Paging_v890_IEs   p_Paging_v890_IEs,

                                          EUTRA_RRC_STATE_Type p_UE_State := RRC_IDLE)

    runs on EUTRA_PTC

 {

    var SubFrameTimingList_Type v_PoList;

    v_PoList := f_EUTRA_PO_List_eDRX (p_CellId, p_PTWLength_Index, p_EDRXValue_Index);

    f_EUTRA_UE_PageSysinfoMod( p_CellId, v_PoList[0], p_UE_State, -, -, p_Paging_v890_IEs, -, true);

 }  

